




Le dispositif utilisé est constitué d'un capteur de force 1kg à jauge de contrainte et pont de Wheatstone.

4 Jauges de contraintes de résistance $1k\Omega$ sont branchées en pont de Wheatstone. A l'équilibre (sans déformation) la tension V_s est nulle. Lorsque les jauges 1 et 4 sont soumises à une déformation, la variation de résistance de R_{j1} et R_{j4} que cela entraîne provoque un déséquilibre du pont. La tension V_s est alors non nulle.



La variation de résistance des jauges de contrainte étant très faible, la valeur de la tension V_s de sortie du pont ne peut pas être mesurée directement en utilisant les entrées analogiques de la carte arduino uno. Le module HC711 est un amplificateur / convertisseur analogique digital 24 bit haute précision. Le HX711 utilise une interface 2 fils pour l'utilisation avec tout type de carte à microcontrôleur disposant d'entrées/sorties digitales. Ce module fonctionne via la librairie arduino HX711.h.

Dispositif complet :



Programme :

Installer la librairie « Grove_LCD_RGB_Backlight-master » pour l'utilisation de l'afficheur et la librairie de l'interface HX711ADC.

Liens pour les librairies.

<https://www.arduino-libraries.info/libraries/hx711-arduino-library>

<http://www.wiki.coworking-aurillac.fr/installation-dun-afficheur-lcd-i2c-grove/>

```
#include "HX711.h"  
#include <Wire.h>  
#include <rgb_lcd.h>
```

```
rgb_lcd lcd;  
const int colorR = 255;  
const int colorG = 255;  
const int colorB = 255;
```

```
const int LOADCELL_DOUT_PIN = 3;  
const int LOADCELL_SCK_PIN = 2;
```

```

float masse;
float tare = 124.01;    // à ajuster lors de l'étalonnage.
HX711 scale;

const int numReadings = 10; // moyenne sur 10 mesures (modifiable)
long valeurs[numReadings];
int readIndex = 0;
long total = 0;
float average = 0;

void setup() {

  lcd.begin (16,2);
  Serial.begin(57600);

  for (int thisReading = 0; thisReading < numReadings; thisReading++)
    valeurs[thisReading] = 0;

  scale.begin(LOADCELL_DOUT_PIN, LOADCELL_SCK_PIN);
}

void loop() {

  if (scale.is_ready()) {

    total = total - valeurs[readIndex];
    valeurs[readIndex] = scale.read();
    total = total + valeurs[readIndex];
    readIndex = readIndex + 1;
  }

  if (readIndex >= numReadings){
  readIndex = 0;
  average = total / numReadings;

  masse = ((0.0004426 *(average)) + tare);    // 0.0004426 a ajuster lors de l'étalonnage

  Serial.print("masse = ");
  Serial.print(masse,2);
  Serial.println("g");

  lcd.clear();
  lcd.setCursor (1,0);
  lcd.print ("MASSE (g)");
  lcd.setCursor (0,1);
  lcd.print (masse,2);

  }
}

```

lien : (Capteur de force et module HX711)
<https://fr.aliexpress.com/item/33046037411.html>